Appendix. STATA Code for Estimating the Mixture Distribution with User Inputs

```
capture log close

*user-defined default directory:  users should define their own
cd "c:/defaultdirectory"

log using "mortality bootstrap.log", replace

clear

*******************************************************************************
*This STATA program simulates a mixture distribution of VSL estimates
*from meta-analyses of US studies.
*If using this code, please cite the companion paper, "The Value of Statistical
*Life: A Meta-Analysis of Meta-Analyses," by H. Spencer Banzhaf,
*Journal of Benefit-Cost Analysis, forthcoming (as of 2022).
*The code may be cut-and-pasted into STATA from this file.
*Alternatively, a do-file can be obtained from the author's webpage at
*http://hsbanzhaf.gsucreate.org/Banzhaf_Research.html or requested by
*contacting him at hsbanzhaf@gsu.edu

*******************************************************************************
*set basic parameters for simulation
set seed 123456789
set obs  1000000
gen str40 VSL_study = ""
gen VSL = .
gen u = .

*******************************************************************************
*set parameters for inflation and income adjustments

*price index for desired year.  Default is 2019 using US CPI-U.
local pindex =      255.657

*income elasticity adjustment.
*0 implies no income adjustment; 1 proportionate adjustment.  Default is 1.
local incelasticity = 1

*income index for desired year.  Default is 2019 US median household income
*from FRED table MEHOINUSA672N (St. Louis Federal Reserve Bank).
*Values for individual studies are for meta-study's year
local incindex = 69560

*******************************************************************************
*Input data for studies being used here.
*These parameters can be changed by the user.
*Default assumes 12 studies.  If more studies are used, change <numstudies>
*and add study details using the format below.
*(Note studies can be eliminated by just setting their weight to zero.)
*For any study, the weights and values can be edited.
*Valid distributions are:
*  normal  ( with parameters Mean, SD),
*  uniform (LB, UB),
*  triangular (a, b, c),
*  or Weibull (Shape, Scale) in dollars [not millions of dollars]

*set number of studies used
local numstudies = 12

*Study 1
```

```
local i          = 1
local study`i'   = "EPA (1997), from 2022 Benmap User's Manual"
local weight`i'  = 0
local distrib`i' = "Weibull"
local Shape`i'   = 1.509588
local Scale`i'   = 9648168
*Base year is 2015 for inflation adjustment, but
*1990 for income adjustment (Benmap 2022)
local pindex`i'  = 237.017
local incindex`i'= 57677


*Study 2
local i          = 2
local study`i'   = "Mrozek & Taylor 2002 NIOSH"
local weight`i'  = (1/6)*(1/2)
local distrib`i' = "normal"
local Mean`i'    = 4.13
local SD`i'      = 1.68
*Base year is 1998
local pindex`i'  = 163.0
local incindex`i'= 61891

display "Mean of `study`i'' in $2019 is  " ///
     `Mean`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'
display "  SD of `study`i'' in $2019 is  " ///
       `SD`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'


*Study 3
local i          = 3
local study`i'   = "Mrozek & Taylor 2002 BLS"
local weight`i'  = (1/6)*(1/2)
local distrib`i' = "normal"
local Mean`i'    = 2.36
local SD`i'      = 0.80
*Base year is 1998
local pindex`i'  = 163.0
local incindex`i'= 60040

display "Mean of `study`i'' in $2019 is  " ///
     `Mean`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'
display "  SD of `study`i'' in $2019 is  " ///
       `SD`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'


*Study 4
local i          = 4
local study`i'   = "Viscusi & Aldy 2003"
local weight`i'  = (1/6)
local distrib`i' = "normal"
local Mean`i'    = 6.1
local SD`i'      = ((8.2-4.6)/3.92)   /*from reported confidence interval*/
*Base year is 2000
local pindex`i'  = 172.2
local incindex`i'= 63292

display "Mean of `study`i'' in $2019 is  " ///
     `Mean`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'
display "  SD of `study`i'' in $2019 is  " ///
       `SD`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'
```

```
*Study 5
local i          = 5
local study`i'   = "Kochi et al. 2006 imputed SEs"
local weight`i'  = (1/6)*(1/2)
local distrib`i' = "normal"
local Mean`i'    = 4.7*(4.1/5.4)
local SD`i'      = 2.3*(4.1/5.4)
*The ratio 4.1/5.4 is the adjustment for excluding studies with negative values,
*from rows 1 and 18 of Table II
*Base year is 2000
local pindex`i'  = 172.2
local incindex`i'= 63292

display "Mean of `study`i'' in $2019 is  " ///
     `Mean`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'
display "  SD of `study`i'' in $2019 is  " ///
       `SD`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'

*Study 6
local i          = 6
local study`i'   = "Kochi et al. 2006 US-only wage hedonics"
local weight`i'  = (1/6)*(1/2)
local distrib`i' = "normal"
local Mean`i'    = (0.618*2.8 + (1-0.618)*8.9)*(4.1/5.4)
local SD`i'      = (0.618*1.3 + (1-0.618)*5.3)*(4.1/5.4)
*0.618 is weight on SP vs hedonic studies.
*It is p solving p*2.8 + (1-p)*9.6 = 5.4 from 1st 3 rows of Table II
*Base year is 2000
local pindex`i'  = 172.2
local incindex`i'= 63292

display "Mean of `study`i'' in $2019 is  " ///
     `Mean`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'
display "  SD of `study`i'' in $2019 is  " ///
       `SD`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'


*Study 7
local i          = 7
local study`i'   = "Viscusi 2018 wide set"
local weight`i'  = (2/6)*(1/4)
local distrib`i' = "normal"
local Mean`i'    = 6.187
local SD`i'      = 0.451
*Base year is 2015
local pindex`i'  = 237.017
local incindex`i'= 61748

display "Mean of `study`i'' in $2019 is  " ///
     `Mean`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'
display "  SD of `study`i'' in $2019 is  " ///
       `SD`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'


*Study 8
local i          = 8
local study`i'   = "Viscusi 2018 best set, best practice"
local weight`i'  = (2/6)*(1/4)
local distrib`i' = "normal"
local Mean`i'    = 4.396
local SD`i'      = 1.792
*Base year is 2015
local pindex`i'  = 237.017
```

```
local incindex`i'= 61748

display "Mean of `study`i'' in $2019 is  " ///
     `Mean`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'
display "  SD of `study`i'' in $2019 is  " ///
       `SD`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'


*Study 9
local i          = 9
local study`i'   = "Viscusi 2018 best set"
local weight`i'  = (2/6)*(1/4)
local distrib`i' = "normal"
local Mean`i'    = 3.641
local SD`i'      = 1.283
*Base year is 2015
local pindex`i'  = 237.017
local incindex`i'= 61748

display "Mean of `study`i'' in $2019 is  " ///
     `Mean`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'
display "  SD of `study`i'' in $2019 is  " ///
       `SD`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'


*Study 10
local i          = 10
local study`i'   = "Viscusi 2018 wide set, best practice"
local weight`i'  = (2/6)*(1/4)
local distrib`i' = "normal"
local Mean`i'    = 11.38
local SD`i'      = 0.288
*Base year is 2015
local pindex`i'  = 237.017
local incindex`i'= 61748

display "Mean of `study`i'' in $2019 is  " ///
     `Mean`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'
display "  SD of `study`i'' in $2019 is  " ///
       `SD`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'


*Study 11
local i          = 11
local study`i'   = "Robinson and Hammitt 2016"
local weight`i'  = (1/6)
local distrib`i' = "triangular"
local a`i'       = 4.2
local b`i'       = 13.7
local c`i'       = 6.4
*c is calibrated to reproduce the mean of the underlying studies, 8.1
*ie solve for c:  8.1 = (4.2 + 13.7 + c)/3 --> c = 6.4
*Base year is 2013
local pindex`i'  = 232.957
local incindex`i'= 59640

display "a of `study`i'' in $2019 is  " ///
        `a`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'
display "b of `study`i'' in $2019 is  " ///
        `b`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'
display "c of `study`i'' in $2019 is  " ///
        `c`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'
```

```
*Study 12
local i          = 12
local study`i'   = "DOT 2016"
local weight`i'  = 0
local distrib`i' = "triangular"
local a`i'       = 5.4
local b`i'       = 13.4
local c`i'       = 8.5
*c is calibrated to reproduce DOT's central value, 9.1
*ie solve for c:  9.1 = (5.4 + 13.4 + c)/3 --> c = 8.5
*Base year is 2012
local pindex`i'  = 229.594
local incindex`i'= 57623

display "a of `study`i'' in $2019 is  " ///
        `a`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'
display "b of `study`i'' in $2019 is  " ///
        `b`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'
display "c of `study`i'' in $2019 is  " ///
        `c`i''*(`pindex'/`pindex`i'')*(`incindex'/`incindex`i'')^`incelasticity'


*****************************************************************************
*renormalize weights so add to 1, if necessary
local double sumweight = 0
forvalues i = 1(1)`numstudies' {
local sumweight = `sumweight' + `weight`i''
}
local sumweight = round(`sumweight', .0000001)
if `sumweight' ~= 1 {
display as red "NB:  WEIGHTS SUM TO `sumweight' AND WERE RENORMALIZED TO ADD TO ONE"
}
forvalues i = 1(1)`numstudies' {
local weight`i' = `weight`i''/`sumweight'
}

*****************************************************************************
*****************************************************************************
*****************************************************************************
/*get VSL values*/

local cumweight = 0

forvalues i = 1(1)`numstudies' {

qui replace VSL_study = "`study`i'"          ///
            if _n > `cumweight'*_N & _n <= (`cumweight' + `weight`i'')*_N

if "`distrib`i'"=="normal" {
qui replace VSL = rnormal(`Mean`i'', `SD`i'') ///
            if _n > `cumweight'*_N & _n <= (`cumweight' + `weight`i'')*_N
}

else if "`distrib`i'"=="uniform" {
qui replace VSL = `LB`i'' + (`UB`i'' - `LB`i'')*uniform() ///
            if _n > `cumweight'*_N & _n <= (`cumweight' + `weight`i'')*_N
}

else if "`distrib`i'"=="triangular" {
qui replace u = uniform() ///
            if _n > `cumweight'*_N & _n <= (`cumweight' + `weight`i'')*_N
```

```
qui replace VSL = ( `a`i'' + sqrt(    u *(`b`i''-`a`i'')*(`c`i''-`a`i'')) )* ///
                  ( u < (`c`i''-`a`i'')/(`b`i''-`a`i'')) ///
               + ( `b`i'' - sqrt( (1-u)*(`b`i''-`a`i'')*(`b`i''-`c`i'')) )* ///
                           ( u >=(`c`i''-`a`i'')/(`b`i''-`a`i'')) ///
            if _n > `cumweight'*_N & _n <= (`cumweight' + `weight`i'')*_N
}

else if "`distrib`i''"=="Weibull" {
qui replace VSL = rweibull(`Shape`i'',`Scale`i'')/1000000 /*convert to mil*/ ///
            if _n > `cumweight'*_N & _n <= (`cumweight' + `weight`i'')*_N
}

qui replace VSL = VSL * (`pindex'/`pindex`i'') * ///
                  ( 1 + `incelasticity'*((`incindex'/`incindex`i'')-1) ) ///
            if _n > `cumweight'*_N & _n <= (`cumweight' + `weight`i'')*_N

local cumweight = `cumweight' + `weight`i''

}

sum VSL, d
local locmean = r(mean)
local sd = r(sd)

twoway kdensity VSL, bw(0.20) n(500) ///
              xtitle("VSL (millions of $2019)") ytitle(Density) ///
              xline(`locmean') graphregion(color(white)) ///
                      note("") title("") legend(off) ///
                      range(-10 30) ///
                      name(gr1, replace) ///
|| (function y=normalden(x,`locmean',`sd'), range(-10 30) lpatter(dash))
graph save gr1 "VSL_density.gph", replace
graph export   "VSL_density.tif", as(tif) replace

capture log close
```